Application Note 300

HIGH PERFORMANCE SEMI-AUTOMATIC TRANSCEIVER TESTING

HEWLETT PACKARD

The Test Set

Modern communications systems, with their ever increasing capabilities, demand sophisticated, high performance test equipment to ensure that they are operating correctly. This Application Note describes a transceiver test set which provides that performance with flexible, easy-to-use standard instrumentation. This test set, the Hewlett-Packard 8903A-E85, makes most in-channel transceiver measurements, either automatically under control of the HP-85F instrument controller or manually using keyboard entry. Table 1 summarizes the test set's capabilities.

In automatic operation, the instruments use keystroke equivalent programming. Each key on the front panel has an equivalent one or two character program code. Thus, test procedures can be developed manually and then translated to the controller's BASIC language by simply substituting program codes for keystrokes. Figure 1 is an example of this. In addition, front panel displays and annunciators for all functions minimize guesswork. This greatly simplifies test program development.

The information contained in this Application Note will help you to assemble this transceiver test set and get it operating. A tape cartridge, the 11723A Application Pac, contains the starter program which is listed in the last section of this note. Most of this program consists of subroutines which control the instruments and make measurements. These subroutines may be used as a starting point for developing your own software.

Table 1. 8903A-E85 abbreviated measurement capabilities.

Transmitter Tests	Receiver Tests	General Tests
Power	Sensitivity	AC and DC level
Frequency	Audio power	Frequency
Frequency error	Audio distortion	Distortion
AM	Signal-to-noise	
FM	SINAD	
Squelch frequency	Quieting	
Squelch deviation	Audio flatness	
Residual and incidental AM and FM		
Microphone sensitivity		
Modulation limiting		
Audio distortion		
Hum and noise		
Audio flatness		



Figure 1. Keystroke-equivalent programming. This example sets the 8656A frequency to 65 MHz.

More information about the instruments used in this test set may be found in the Operating and Service Manuals for each unit. Application Note 286-1, Applications and Operation of the 8901A Modulation Analyzer, describes transmitter testing in more detail. Information about programming the HP-85 is contained in the Owner's Manual and the I/O Programming Guide, which are supplied with the unit.

Equipment and Setup

The following instruments and accessories are necessary to assemble the test set. They may be ordered as a set from Hewlett-Packard under the number 8903A-E85.

8656A Signal Generator,
  Opt. 002, rear panel connections
8901A Modulation Analyzer,
  Opt. 001, rear panel connections
  Opt. 002, high stability time base
8903A Audio Analyzer,
  Opt. 001, rear panel connections
8903A-K85 Switching Module
59306A Relay Actuator
8498A Opt. 030 Attenuator, 30 dB 25 watt
85F Computer System, HP-IB interface
11723A Application Pac program tape
10833A Low-RFI HP-IB cable, 1 m (3 required)
11035A BNC-banana cable, 30 cm
11170A BNC cable, 30 cm
11170B BNC cable, 60 cm (2 required)
11170C BNC cable, 1.2 m (2 required)
11500B N cable, 60 cm (2 required)

Rear-panel connector options for the instruments are not required for proper operation. However, if the instruments have front-panel connectors, some of the cables will need to be longer.

Figure 2 is the block diagram of the test set. The signal paths in the test set are controlled by the 8903A-K85 Switching Module. This device has two modes of operation, transmit and receive. They are selected with an external contact closure. This closure is provided by the 59306A Relay Actuator. Figures 3A and 3B show the instrument configurations for testing a receiver or transmitter.

To connect the equipment as shown in Figure 2, follow these steps:

1) Turn off ac power to all instruments.

2) Insert the I/O ROM and HP-IB interface supplied with the controller into two of the slots in the unit's rear panel.

3) Using the free end of the HP-IB interface cable and the three 10833A cables, connect the signal generator, audio analyzer, modulation analyzer, and relay actuator to the HP-IB interface.

4) Using the two 11170C 1.2 m BNC cables, connect the audio analyzer Input and Output to the corresponding connectors on the switching module. Set the **FLOAT/GND**

switches on the audio analyzer front panel to the **GND** position.

5) Using the 11170A 30 cm BNC cable, connect together the modulation analyzer Time Base Output and the signal generator Time Base Input.

6) Using the 11035A BNC-banana cable, connect the switching module Control connector to the 6A and 6C binding posts on the relay actuator.

7) Using the remaining cables, connect the signal generator RF Output and Modulation Input and the modulation analyzer RF Input and Modulation Output to the corresponding connectors on the switching module.

8) Attach the 30 dB attenuator to the RF connector on the front of the switching module.

9) Connect the transceiver to the connectors on the front panel of the switching module. If the transceiver requires an external audio load resistor, connect it in parallel with the SPKR IN port on the switching module.

10) Turn on all instruments and the unit to be tested.

HP-IB Addresses

Table 2 lists the HP-IB addresses for the instruments in the test set. These addresses, which are set at the factory before shipment, are required for proper operation of the 11723A Application Pac software. To display the addresses for each instrument:

8901A: key in **21. SPCL.** The display, which is in binary, should be 01110.000 (decimal 14).

8903A: key in **21.1 SPCL.** The display, which is in decimal, should be 28.

8656A: press **HP-IB ADRS** key. The display, which is in decimal, should be 07.

59306A: the address switches on the rear panel should be set as shown in Figure 4.

The procedure for changing the address of the 8901A, 8903A, or 8656A requires resetting switches inside the instrument. Refer to Section 2 of the appropriate Operating and Service Manual for details.



Figure 2. 8903A-E85 Transceiver Test Set block diagram.



Figure 3A. Receive configuration.



Figure 3B. Transmit configuration.

Verification Procedure

Before attempting to operate the test set automatically, the connections should be verified. To do this, with a known good transceiver follow these steps manually:

1) Set button 6 on the relay actuator to the out (off) position. This selects the signal paths in the switching module used for receiver tests.

2) Set the signal generator as follows:
Frequency: receiver frequency
Modulation: external, select test deviation/depth
Output level: −17 dBm

3) Set the audio analyzer as follows:
Source frequency: 1 kHz or as desired
Source level: +3 dBV

4) Verify that the signal generator EXT HI and EXT LO indicators are off. If one is on, check the audio source level and the connections between the audio analyzer Output, the switching module, and the signal generator Modulation Input.

5) Press the **AC LEVEL** key on the audio analyzer. The display should now be indicating the audio output voltage from the receiver. The reading should be stable and it should change as you vary the receiver volume control. If not, check the connections between the signal generator Output, the switching module, and the transceiver antenna connector. Also check the connections between the receiver audio output, the switching module, and the audio analyzer input.

6) Set button 6 on the relay actuator to the in (on) position. This keys the transmitter and selects the transmitter signal paths in the switching module.

7) Verify that the transmitter has been keyed. If not, check the connection between the relay actuator and the switching module, and the transmitter key connection to the switching module. Also check that ac power has been applied to the switching module.

8) Press the **RF LEVEL** key on the modulation analyzer. The display should show the transmitter output power, divided by 1000 (because of the 30 dB attenuator). If not, check the connection between the modulation analyzer Input and the switching module.

9) Select the AM or FM mode of the modulation analyzer, as appropriate for the transceiver. The display should show the peak modulation in percent or kHz. The audio source output level is high for most transceivers, so you will probably observe the limiting value. If you see only residual modulation, check the connection between the switching module and the transmitter audio input.

10) Press the **DISTN** key on the audio analyzer. The display should show the transmitter audio distortion. If the display shows no signal (two dashes), check the connection between the modulation analyzer Modulation Output and the switching module.

This completes the verification of interconnections for the test set.

Table 2. HP-IB addresses.

Instrument	Address	
	Decimal	Binary
8901A	14	01110
8903A	28	11100
8656A	07	00111
59306A	16	10000

Figure 4. 59306 address switch settings.

Operating the 11723A Application Pac Software

The program contained on the 11723A tape cartridge is a general-purpose test program which makes fourteen common tests for AM and FM transceivers. Figure 5 is an example of the test package printout. To run this program, insert the tape cartridge into the HP-85 controller and turn the controller on. The program should automatically load and run.

The first time the program runs after the controller is turned on, the controller's internal clock must be set. The program will ask for the current date and time. After each entry, press **END LINE.** The program then asks several questions about the transceiver test conditions. Again, press **END LINE** after each entry.

When the test conditions have all been entered, the calculator will display a heading, the name of the operator and the radio identification. If you want to make a hard copy from the display before going on with the test, press (shift) **COPY.** Anything in the display may be copied by doing this.

To start the transmitter test, press **CONT**inue. The test results will appear on the screen as the tests are completed. When all of the tests are complete, the calculator will beep. Press **CONT**inue again to go on to the receiver test. Before

testing the receiver, the program will prompt you to adjust the receiver volume control for the rated audio power. Adjust the volume control until the blinking arrow is in the bright center portion of the display. After this is done, the receiver tests will begin. After these tests are finished, press **CONT**inue once more to see the receiver audio response. As each of the tests is in progress, you can watch the instruments and see the results appear in the displays. This feature is very helpful when developing new test procedures.

After the tests are all finished, you may press **CONT**inue again to print the total elapsed test time. This is the time that the test set actually spent testing the transceiver, and does not include waiting time between tests. To repeat the entire test using the same test conditions, just press **CONT**inue again.

Table 3 is a summary of the keys and commands which are useful with the test set.

Table 3. Test set command summary

Key/Command	Action
RUN	Restarts program execution
END LINE	Terminates any data entry
CONT	Go on to next part of test; restart test if finished
(shift)**COPY**	Copies CRT screen to internal printer
CONT 7100	Restarts transmitter test*
CONT 7200	Restarts receiver test*
CONT 7300	Restarts audio response test*

*Press **END LINE** after typing in these commands. The test conditions must have been entered before using these commands or an error will result.

Software Explanation

The 11723A Application Pac program is a comprehensive demonstration/starter program, which illustrates how to use the instruments with a controller. It is capable of testing most AM and FM communications transceivers in its basic form, and it may be expanded to do any test which the instruments can do manually. The program has a modular structure so that it is easily modified and customized for specific applications. The structure is outlined in Figure 6, and further explained in the following section, which also contains a brief description of each module and how to use it. The annotated program listing at the end of this note shows how these modules have been used to create the 11723A Application Pac test package. You may use these modules in a similar fashion to write your own test package.



Figure 5. 8903A-E85 test set printout.

The software modules consist of a main program (starting at line 7000) and three types of subroutines: instrument subroutines, measurement subroutines, and utility subroutines. The instrument subroutines are used to control the instruments themselves and to make many of the measurements. For example, there are subroutines to set the signal generator RF frequency and to measure audio distortion. Many transceiver measurements are made directly by the instruments. For these measurements, instrument subroutines are all that is required.

More complex measurements like usable sensitivity, in which the receiver output SINAD or S/N ratio is measured as the signal generator level is varied, are made using measurement subroutines. These subroutines call the instrument subroutines and have additional computation and logic (IF. . .THEN) statements included to control the measurement process. Because the instruments themselves have internal microprocessors which do much of what is required to test a radio, there are only five measurement subroutines in the program.



Figure 6. 11723A Application Pac software structure.

The utility subroutines are used to perform overhead and control tasks. They initialize the instruments, control the 8903A-K85 Switching Module, and provide an elapsed time function. These subroutines are most often called at the beginning and end of test groups to configure the test set for the desired measurements.

The main program contains statements which call the subroutines and display the results. The main program also does unit conversions where necessary, and it provides the subroutines with the information needed to run the tests, such as test conditions and initial estimates. The main program also has four subroutines which create display graphics. These are located at lines numbered 9000 and above.

The operating manuals provided with the controller and the instruments contain more information about the program codes and protocols required for HP-IB operation. Many of the program codes also appear on the instruments' front panels, near the corresponding keys used in manual operation.

Subroutine Descriptions

This section contains a description of each of the subroutines included in the 11723A Application Pac. The information included here is helpful when using the subroutines to build test packages and when modifying the 11723A Application Pac program.

Function, **Calling Syntax,** *Condition/Variables/Description*

8656A Signal Generator

RF level **GOSUB 1100** *A = level, dBm. K = antenna attenuator, dB. Sets RF level at attenuator output.*

RF frequency **GOSUB 1200** *A = frequency, MHz.*

FM deviation **GOSUB 1300** *A = deviation, kHz.*

AM depth **GOSUB 1400** *A = depth, percent.*

8901A Modulation Analyzer

AM depth **GOSUB 2100** *Sets A to depth in percent.*

Count freq. **GOSUB 2200** *Sets A to input frequency in Hz.*

De-emphasis **GOSUB 2300** *A = de-emphasis, μs. B = 0 for pre-display off, 1 for pre-display on.*

Detector **GOSUB 2400** *A = 1 for peak+, 2 for peak−, 4 for average. B = 0 for peak hold off, 1 for peak hold on.*

Filters **GOSUB 2500** *A = high pass frequency, Hz (0 = none). B = low pass frequency, kHz (0 = none).*

FM deviation **GOSUB 2600** *Sets A to deviation in Hz.*

Power **GOSUB 2700** *Sets A to peak envelope power in watts.*

Data **GOSUB 2800** *This subroutine makes another measurement in the current mode and puts the result in A.*

Fix mod range **GOSUB 2900** *This subroutine fixes the modulation range to the current value, suppressing autoranging.*

8903A Audio Source

Frequency **GOSUB 3100** *A = frequency, Hz.*
Level **GOSUB 3200** *A = level, dBV.*
Source only **GOSUB 3300** *Displays source settings.*
Source off **GOSUB 3400** *Sets source amplitude to 0.*

8903A Audio Analyzer

AC volts **GOSUB 4100** *Sets A to rms input voltage.*
Distortion **GOSUB 4200** *Sets A to input distortion in percent.*
SINAD **GOSUB 4300** *Sets A to SINAD in dB.*
S/N **GOSUB 4400** *Sets A to signal-to-noise ratio in dB. The audio analyzer source must drive the device being tested.*
Watts **GOSUB 4500** *Z = load impedance, ohms. Sets A to equivalent input power in watts.*
Data **GOSUB 4600** *This subroutine makes another measurement in the current mode (except frequency count) and puts the result into A.*
Freq. count **GOSUB 4700** *Sets A to input frequency in Hz.*
HP/BP filters **GOSUB 4800** *A = 0 for no filter, 1 for 400 Hz HP, 2 for psophometric.*
DC volts **GOSUB 4900** *Sets A to dc input voltage.*

Utility

Initialize **GOSUB 5100** *Initializes the instruments and HP-IB interface for proper test set operation.*
Transmit **GOSUB 5200** *Selects the transmitter signal paths in the 8903A-K85 switching module.*
Receive **GOSUB 5300** *Selects the receiver signal paths in the 8903A-K85 switching module.*
Timeout **GOSUB 5400** *Prints diagnostic message and stops program execution. Returns HP-IB to local mode.*
ET Clear **GOSUB 5500** *Sets elapsed timer to 0.*
ET Start **GOSUB 5600** *Starts elapsed timer.*
ET Stop **GOSUB 5700** *Stops elapsed timer. Elapsed time is available in the variable T1.*

Measurement Subroutines

Squelch Test **GOSUB 6100** *Checks transmitter for the presence of a subaudible squelch tone. If a tone is present, Q is set to 1, Q0 is set to the tone frequency, Q1 is set to the squelch deviation, the modulation analyzer filters are set to 300 Hz and 15 kHz, and the audio analyzer 400 Hz filter is selected. If the squelch tone is absent, Q is set to 0 and the modulation analyzer filters are set to 50 Hz and 15 kHz.*
Microphone Sensitivity **GOSUB 6200** *A = Initial sensitivity estimate, dBV from 600 ohms. B = Audio source frequency, kHz. M = Desired modulation, percent or Hz. U = 1 for AM, 2 for FM. Sets A to the audio level, in dBV from 600 ohms, which will deliver the modulation specified by M.*
Modulation Limiting **GOSUB 6300** *A = Microphone sensitivity, dBV from 600 ohms. U = 1 for AM, 2 for FM. Sets B to peak limiting in Hz or percent. Sets C to steady-state limiting. Sets transmitter for rated modulation.*
Receiver Sensitivity **GOSUB 6500** *A = estimated sensitivity, dBm. U = 1 for AM, 2 for FM. Y = 1 for US/EIA, 2 for Europe/CEPT. P = rated receiver audio power, watts. Z = audio load impedance, ohms. Calls subroutine 6600, receiver set. Sets A to level in dBm required to produce 12 or 20 dB SINAD (FM) or 10 dB S/N (AM).*
Receiver Set **GOSUB 6600** *R = receiver channel frequency, MHz. U = 1 for AM, 2 for FM. P = rated receiver audio power, watts. Z = audio load impedance, ohms. M = modulation desired, percent or Hz. Q2 = desired squelch tone frequency, Hz (0 for none). Sets B to actual audio power set by operator. Uses 1 kHz audio tone. Sets the instruments as specified by R, U, P, Z, M, and Q2 for receiver tests.*

Program Variables

Table 4 is a summary of the variables to which the subroutines refer as they make their measurements. These variables may be set in any convenient fashion. In the 11723A Application Pac program, most of them are specified by the operator in response to the prompts generated in the program lines numbered 7000 through 7099.

Table 4. Program variables.

R	Receiver frequency, MHz
T	Transmitter frequency, MHz
P	Receiver rated audio power, watts
Z	Receiver audio load impedance, ohms, (maximum 999)
U	Set to 1 for AM transceivers, 2 for FM
M	Test modulation, percent (AM) or Hz (FM) (used for microphone sensitivity and all receiver tests)
V*	De-emphasis, microseconds
K	Antenna pad attenuation, dB
Q2*	Receiver squelch tone frequency (0 for none); deviation will be 33% of M
Y	Set to 1 for EIA standards, 2 for CEPT
T1	Elapsed test time, seconds

*Used for FM transceivers only. Must be set to 0 for AM transceivers.

In addition, the following variables are used by some of the subroutines for temporary storage: A, B, C, D, E, A1, B1, N.

Q, Q0, and Q1 are set by the squelch test routine.

T0 is used by the elapsed timer and should not be changed while the timer is in use.

11723A Application Pac — Program Introduction

```
 10 ! ***************************
 15 !
 20 ! 8903A-E85
 25 ! TRANSCEIVER TEST SET
 30 !
 35 ! 11723A APPLICATION PAC
 40 !
 45 ! V01.10150 REV I; WJD
 50 !
 55 ! ***************************
 60 !
 65 GOTO 7000 ! START OF PROGRAM
 70 !
 75 ! ***************************
 80 !
 85 ! SUBROUTINE DIRECTORY
 90 !
100 ! 8656A ********************
110 ! 1100 RF LEVEL
120 ! 1200 RF FREQ
130 ! 1300 FM DEV
140 ! 1400 AM DEPTH
200 ! 8901A ********************
210 ! 2100 AM
220 ! 2200 COUNT
230 ! 2300 DE EMPHASIS
240 ! 2400 DETECTOR
250 ! 2500 FILTERS
260 ! 2600 FM
270 ! 2700 PEP
280 ! 2800 DATA
290 ! 2900 FIX MOD RANGE
300 ! 8903A ********************
310 ! 3100 AF FREQ
320 ! 3200 AF LEVEL
330 ! 3300 SOURCE ONLY
340 ! 3400 SOURCE OFF
410 ! 4100 AC VOLTS
420 ! 4200 DISTORTION
430 ! 4300 SINAD
440 ! 4400 S/N
450 ! 4500 WATTS
460 ! 4600 DATA
470 ! 4700 AF COUNT
480 ! 4800 HP/BP FILTERS
490 ! 4900 DC VOLTS
500 ! UTILITY ******************
510 ! 5100 INIT
520 ! 5200 XMIT
530 ! 5300 RCV
540 ! 5400 TIMEOUT
550 ! 5500 ET CLEAR
560 ! 5600 ET START
570 ! 5700 ET STOP
600 ! MEASUREMENT **************
610 ! 6100 SQUELCHTEST
620 ! 6200 MICSENS
630 ! 6300 MODLIM
650 ! 6500 RXSENS
660 ! 6600 RXSET
699 !
999 IMAGE K
```

Title Block

First main program line is 7000

This section lists the starting line numbers for all of the subroutines in the 11723A Application Pac program. Information about how to use the subroutines is contained in the previous section.

Literal image used for general-purpose output statements. Causes characters to be sent exactly as specified without trailing blanks.

8656A Subroutines

```
1000 ! 8656A INST SUBS
1010 !
1100 ! RF LEVEL(A),dBm
1110 IMAGE "AP",4D.D,"DM"
1120 OUTPUT 707 USING 1110 ; A+K
1130 RETURN
1199 !
1200 ! RF FREQ(A),MHz
1210 IMAGE "FR",3D.5D,"MZ"
1220 OUTPUT 707 USING 1210 ; A
1230 RETURN
1299 !
1300 ! FM DEV(A),kHz
```

Set RF Level in dBm at attenuator output

Image for 8656A amplitude

K = attenuator value in dB (correction)

Set RF frequency in MHz

Image for 8656A frequency

Set FM deviation in kHz

```
1310 IMAGE "FM",2D.D,"KZ"
1320 OUTPUT 707 USING 1310 ; A
1330 RETURN
1399 !
1400 ! AM DEPTH(A),%
1410 IMAGE "AM",DD.D,"PC"
1420 OUTPUT 707 USING 1410 ; A
1430 RETURN
1499 !
```

Image for 8656A deviation. Does not specify modulation source

Set AM depth in percent

Image for 8656A depth. Does not specify modulation source.

8901A Subroutines

```
2000 ! 8901A INST SUBS
2010 !
2100 ! AM(A),%
2110 OUTPUT 714 USING 999 ; "M1T
     3"
2120 ENTER 714 ; A
2130 RETURN
2199 !
2200 ! CNT FRQ(A),Hz
2210 OUTPUT 714 USING 999 ; "M5A
     U7.1SP4.1SPT3"
2220 ENTER 714 ; A
2230 OUTPUT 714 USING 999 ; "AU"
2240 IF (A-9000000000)/100#10 TH
     EN RETURN
2250 OUTPUT 714 USING 999 ; "T3"
2260 ENTER 714 ; A
2270 RETURN
2299 !
2300 ! FM DEEMPH(A,B),uS
2310 A1=MIN(A,100)
2320 IMAGE "P",D,"P",D
2330 OUTPUT 714 USING 2320 ; B#0
     ,A1/25+(A1>=25)
2340 RETURN
2399 !
2400 ! DET(A,B)
2410 IMAGE "D",D
2420 OUTPUT 714 USING 2410 ; A
2430 IF B=0 THEN RETURN
2440 OUTPUT 714 USING 999 ; "D3"
2450 RETURN
2499 !
2500 ! FLT(A,B)
2510 A1=(A>0)+(A>50)
2520 B1=(B>0)+(B>3)+(B>15)
2530 IMAGE "H",D,"L",D
2540 OUTPUT 714 USING 2530 ; A1,
     B1
2550 RETURN
2599 !
2600 ! FM(A),Hz
2610 OUTPUT 714 USING 999 ; "M2T
     3"
2620 ENTER 714 USING 999 ; A
2630 RETURN
2699 !
2700 ! PK ENV PWR(A),WATTS
2710 OUTPUT 714 USING 999 ; "M4T
     3"
2720 ENTER 714 ; A
2730 RETURN
2799 !
2800 ! DATA(A)
2810 OUTPUT 714 USING 999 ; "T3"
2820 ENTER 714 ; A
2830 RETURN
2899 !
2900 ! FIX MOD RANGE
2910 OUTPUT 714 USING 999 ; "SST
     2" ! DISP CURRENT SETTINGS
2920 ENTER 714 USING "XX,D" ; A
2930 IMAGE "2.",D,"SP"
2940 OUTPUT 714 USING 2930 ; A
2950 RETURN
2999 !
```

Measure AM depth

M1 = AM; T3 = Trigger with Settling

A will be in percent

Measure input frequency

M5 = Frequency; 7.1 SP = 10 Hz Resolution; 4.1 SP = Track Mode; T3 = Trigger

A will be in Hz

Clear special functions

Possible error: if input frequency >1 GHz, cannot use 10 Hz resolution (too many digits)

If error, make another reading with 100 Hz resolution

Set FM de-emphasis. A = 25, 50, 75, 750; B = 0, 1

Statements convert value in μs to codes P0 through P5

Select detectors A = 1, 2, 4; B = 0, 1

Converts values to codes D1 through D4

Code D3 if B # 0 (peak hold).

Select filters. A = 0, 50, 300; B = 0, 3, 15, 20

Convert values in Hz, kHz to codes H0—H2 and L0—L3

Measure FM deviation

M2 = FM; T3 = Trigger with settling

A will be in Hz

Measure PEP

M4 = Power; T3 = Trigger with settling

A will be in watts

Make another measurement

Trigger, use same mode (M1—M5) as last measurement

Fix modulation range; suppress auto-ranging

SS = display instrument settings; T2 = trigger

Read only mod. range digit. Ignore others

Image for 8901A mod. range

8903A Subroutines (Source)

```
3000 ! 8903A INST SUBS - SOURCE
3010 !
3100 ! FRQ(A),kHz ---------- Set audio frequency in kHz
3110 IMAGE "FR",DD.4D,"KZ" ---------- Image for 8903A source frequency
3120 OUTPUT 728 USING 3110 ; A
3130 RETURN
3140 !
3200 ! LEV(A),dBv-600Ω ---------- Set audio level, dBV EMF. Note: 8903A source has 600 Ω impedance
3210 IMAGE "AP",3D.DD,"DV" ---------- Image for 8903A source level
3220 OUTPUT 728 USING 3210 ; A
3230 RETURN
3299 !
3300 ! SOURCE ONLY ---------- Display source settings
3310 OUTPUT 728 USING 999 ; "10 ---------- 10.SP = display settings; T3 = trigger
     SPT3"
3320 RETURN
3330 !
3400 ! AF SOURCE OFF ---------- Set audio amplitude to 0 (not possible in dBv)
3410 OUTPUT 728 USING 999 ; "AP0 ---------- AP0MV = amplitude 0 millivolts
     MV"
3420 RETURN
3499 !
```

8903A Subroutines (Analyzer)

```
4000 ! 8903A INST SUBS - MEAS
4010 !
4100 ! ACV(A) ---------- Measure ac rms level
4110 OUTPUT 728 USING 999 ; "M1T ---------- M1 = ac level; T3 = trigger with settling
     3"
4120 ENTER 728 ; A ---------- A will be in volts
4130 RETURN
4140 !
4200 ! DIST(A),% ---------- Measure harmonic distortion, notch method.
4210 OUTPUT 728 USING 999 ; "M3T ---------- M3 = distortion; T3 = trigger with settling
     3"
4220 ENTER 728 ; A ---------- A will be in percent
4230 RETURN
4299 !
4300 ! SINAD(A),dB ---------- Measure SINAD using notch method
4310 OUTPUT 728 USING 999 ; "M2T ---------- M2 = SINAD; T3 = trigger with settling
     3"
4320 ENTER 728 ; A ---------- A will be in dB
4330 RETURN
4399 !
4400 ! S/N(A),dB ---------- Measure S/N ratio, switching 8903A source
4410 OUTPUT 728 USING 999 ; "S2T ---------- S2 = S/N ratio; T3 = Trigger with Settling
     3"
4420 ENTER 728 ; A ---------- A will be in dB
4430 RETURN
4499 !
4500 ! WATTS(A,Z),WATTS,LOADΩ ---------- Measure equivalent power into load = ZΩ
4510 IMAGE "19.",3Z,"SPT3" ---------- 19.NNN SP = watts into NNNΩ
4520 OUTPUT 728 USING 4510 ; Z ---------- Set NNN = value specified by Z
4530 ENTER 728 ; A ---------- A will be in watts
4540 RETURN
4599 !
4600 ! DATA(A) ---------- Make another measurement (except frequency)
4610 OUTPUT 728 USING 999 ; "T3" ---------- T3 = trigger with settling
4620 ENTER 728 ; A ---------- A will be in same units
4630 RETURN
4699 !
4700 ! AF CNT(A),Hz ---------- Count audio frequency
4710 OUTPUT 728 USING 999 ; "M1R ---------- M1 = ac volts (fastest mode); RL = left display (frequency)
     LT3"
4720 ENTER 728 ; A ---------- A will be in Hz
4730 OUTPUT 728 USING 999 ; "RR" ---------- RR = right display
4740 RETURN
4799 !
4800 ! HP/BP(A) ---------- Set HP/BP filters. A =0, 1, 2
4810 IMAGE "H",D ---------- Convert values to codes H0—H2
4820 OUTPUT 728 USING 4810 ; A
4830 RETURN
4899 !
4900 ! DCV(A) ---------- Measure dc level
4910 OUTPUT 728 USING 999 ; "S1T ---------- S1 = dc level; T3 = trigger with settling
     3"
4920 ENTER 728 ; A ---------- A will be in volts
4930 RETURN
4999 !
```

Utility Subroutines

```
5000 ! UTILITY SUBS
5010 !
5100 ! INIT                                                    Initialize interface and instruments
5110 ABORTIO 7
5120 CLEAR 7                                                   HP-IB initialization
5130 REMOTE 7
5140 SET TIMEOUT 7;10000
5145 ON TIMEOUT 7 GOSUB 5400                                   10 second timeout to trap missing instruments
5150 OUTPUT 707 USING 999 ; "S1A                               8656A: select external mod. source
     MFM"
5160 OUTPUT 714 USING 999 ; "CLA                               8901A: initialize detectors, filters, deemphasis, measurement modes
     U7.1SPD1H1L1P0C0R0M2"
5170 OUTPUT 716 USING 999 ; "B12                               59306A/8903A-K85: select receive mode
     3456"
5180 OUTPUT 728 USING 999 ; "16                                8903A; full resolution on SINAD, S/N (16.1 SP); display source settings (10.SP)
     1SP,10 SP"
5190 RETURN
5199 !
5200 ! XMIT
5210 OUTPUT 716 USING 999 ; "A6"                               59036A/8903A-K85: select transmit mode
5220 RETURN
5299 !
5300 ! RCV
5310 OUTPUT 716 USING 999 ; "B6"                               59036A/8903A-K85; select receive mode
5320 RETURN
5399 !
5400 ! TIMEOUT ROUTINE
5410 LOCAL 7                                                   Return instruments to local mode
5420 PRINT "*TIMEOUT ERROR*"                                   Print warning message
5430 STOP                                                      Terminate program
5499 !
5500 ! ET CLEAR                                                Clear elapsed timer function
5510 T1=0                                                      T1 = total elapsed time
5520 RETURN
5599 !
5600 ! ET START
5610 T0=TIME                                                   T0: start interval time
5620 RETURN
5699 !
5700 ! ET STOP
5710 T1=T1+TIME-T0                                             Add time since T0 to T1
5720 T0=TIME                                                   Reset T0 to avoid double-counting
5730 RETURN
5799 !
```

Measurement Subroutines (Transmitter)

```
6100 ! SQUELCHTEST                                             Checks for subaudible squelch tone
6110 A=50 ! HP Hz
6115 B=15 ! LP kHz                                             Set 8901A filters; 50 Hz - 15 kHz BW
6120 GOSUB 2500 ! FLT
6125 GOSUB 2600 ! FM                                           Measure FM
6130 IF A<200 THEN Q=0 @ RETURN                                If less than 200 Hz deviation, no squelch
6135 Q=1 ! SQ ON                                               Squelch tone present
6140 Q1=A ! DEV                                                Q1 = deviation in Hz
6145 GOSUB 4700 ! AF CNT                                       Q0 = squelch frequency in Hz
6150 Q0=A ! SQ FRQ
6155 A=300 ! HP Hz
6160 B=15 ! LP kHz                                             Set 8901A filters: 300 Hz - 15 kHz BW
6165 GOSUB 2500 ! FLT                                          Minimizes effects of squelch on mic. sens. routine
6170 A=1 ! 400Hz HP                                            Set 8903A 400 HP filter for transmitter distortion measurement
6175 GOSUB 4800 ! HP-BP
6180 RETURN
6199 !

6200 ! MICSENS                                                 Finds microphone sensitivity
6205 C=A ! dBv                                                 Temporary storage of initial estimate
6210 A=B ! kHz
6215 GOSUB 3100 ! AFF                                          Set test frequency to B
6220 ON U GOSUB 2100,2600 ! AM,F                               Use either AM or FM, as specified by U
     M
6225 A=C ! dBv                                                 Set level of audio source
6230 GOSUB 3200 ! AFL
6235 GOSUB 2800 ! DATA                                         Make another AM/FM measurement
6240 C=C+20*LGT(M/A) ! NEW EST                                 Estimate new audio level using extrapolation
6245 IF ABS((M-A)/M)<.025 THEN 6                               If error <2.5%, then close enough
     260
6250 IF C>-64 AND C<15 THEN 6225                               If new level not too large or small, try again
6255 PRINT "*MICSENS FAILED*"                                  Otherwise print warning and return
6260 A=C ! SENS dBv                                            A = Best estimate of microphone sensitivity
6265 RETURN
6299 !
```

```
6300 ! MODLIM                                  Measures modulation limiting
6305 GOSUB 3200 ! AFL                         Set audio level for rated modulation (A)
6310 D=A                                      Temporary storage for A
6315 ON U GOSUB 2100,2600 ! AM,F              Use either AM or FM as specified by U
     M
6320 WAIT 500                                 Wait for instruments and UUT to settle
6325 A=1 ! PK +                               Set 8901A detector: peak hold, (+)
6330 B=1 ! HOLD                               
6335 GOSUB 2400 ! DET                         
6340 E=MIN(15.5,D+20)-D                       E = amplitude increment. 20 dB used unless it exceeds 8903A capabilities
6345 IMAGE "AN",DD.DD,"DBUPT2DNT              Increment amplitude up, down, up
     2UP"                                     
6350 OUTPUT 728 USING 6345 ; E                
6355 GOSUB 2800 ! DATA                        Read back peak instantaneous modulation
6360 E=A                                      Store in E (temporary)
6365 A=1 ! PK +                               Turn off peak hold
6370 B=0 ! HOLD OFF                           
6375 GOSUB 2400 ! DET                         
6380 GOSUB 2800 ! DATA                        Measure steady-state modulation
6385 C=A                                      Set: A back to mic. sensitivity; B to instantaneous limiting; C to steady state limiting
6390 B=E                                      
6395 A=D                                      
6400 GOSUB 3200 ! AFL                         Reset audio level for rated modulation
6405 RETURN
6499 !
```

Measurement Subroutines (Receiver)

```
6500 ! RXSENS                                 Receiver Sensitivity
6505 C=A ! EST SENS                           Temporary storage for initial estimate
6510 GOSUB 6600 ! RXSET                       Set instruments for receiver tests
6515 ON U GOSUB 4400,4300 ! SN,S              For AM, measure S/N ratio; FM, measure SINAD
     INAD
6517 IF Y=2 THEN OUTPUT 728 USIN              For CEPT, use P53 psophometric filter (H2)
     G 999 ; "7.1SP,H2" ! PSOPH               Enable meter for >24 dB range (7.1 SP)
     FILT
6520 A=C                                      Set RF level to initial estimate
6525 GOSUB 1100 ! RFL                         
6530 GOSUB 4600 ! DATA                        Measure S/N or SINAD
6535 A=2*U+8*Y-A ! ERROR TERM                 Calculate error in dB from 10, 12, or 20
6540 C=C+A/2 ! NEW EST                        New sensitivity estimate
6545 IF ABS(A)<1 THEN 6565                    If error small, then finished
6550 IF C+K>=-127 AND C+K<13 THE              If new estimate within 8656A range, try again
     N 6520
6555 PRINT "*RXSENS FAILED*"                  Otherwise, print failure message
6560 RETURN
6565 IF U=1 THEN A=C @ RETURN                 AM tests: complete
6570 GOSUB 4500 ! WATTS                       FM tests: measure audio power
6575 IF A>.5*P THEN A=C @ RETURN              If >50% of rated, then finished
6580 C=C+10*LGT(.55*P/A)                      Increment RF level to increase audio power
6585 A=C                                      Set RF level
6590 GOSUB 1100 ! RFL                         
6595 GOTO 6570                                Loop to measure power
6599 !

6600 ! RXSET                                  Set instruments for receiver tests
6602 GOSUB 2700 ! PEP - ELIMINAT              Power mode inserts full attenuation between 8901A mixer and RF port. Reduces LO feedthru
     ES 8901 LO FEEDTHRU
6604 A=R ! MHz                                Set 8656A to receiver channel frequency
6606 GOSUB 1200 ! RFF                         
6608 A=-47 ! dBm                              Set 8656A level to -47 dBm (1000 μV)
6610 GOSUB 1100 ! RFL                         
6612 A=1 ! kHz                                Set 8903A source to 1 kHz
6614 GOSUB 3100 ! AFF                         
6620 A=3 ! dBv                                Set 8903A source to 3 dBV, for rated modulation with 8656A
6625 GOSUB 3200 ! AFL                         
6630 A=M ! MOD %/Hz                           Set 8656A modulation
6635 IF U=2 THEN 6650 ! FM                    
6640 GOSUB 1400 ! RFA                         
6645 GOTO 6690                                
6650 A=A/1000 ! Hz→kHz                        
6652 GOSUB 1300 ! RFD                         
6655 IF NOT Q2 THEN 6690 ! NO SQ              Next six statements generate subaudible squelch tone. Skip if no squelch
     UELCH
6660 OUTPUT 707 USING 999 ; "S3F              Internal 1 kHz mod source on
     M" ! INT 1kHz
6665 OUTPUT 728 USING 999 ; "FR1              Set SINAD notch to 1 kHz; set source to 1 kHz (FR1KZ), SINAD mode (M2T3), freeze
     KZM2T36.1SP" ! FREEZE SINAD              notch frequency (6.1 SP)
      NOTCH @ 1kHz
6670 A=Q2/1000 ! Hz→kHz SQ FRQ                Set source to squelch frequency
6675 GOSUB 3100 ! AFF                         
```

```
6680 A=-6.63 ! 33% OF RATED DEV
     FOR SQUELCH TONE
6685 GOSUB 3200 ! AFL
6690 A=0 ! FILT OFF
6692 GOSUB 4800 ! HP/BP
6694 GOSUB 6700 ! SETLEVEL
6696 RETURN
6699 !
```

Set source level for 33% rated deviation

8903A HP/BP filters off

Set receiver for rated audio power

Measurement Subroutines (Level Set)

```
6700 ! SETLEVEL
6705 DEG @ A1=48
6707 WAIT 500
6710 GOSUB 4500 ! WATTS
6715 IF ABS((A-P)/P)<.05 THEN 67
     75
6720 GOSUB 5700 ! ET STOP
6725 GOSUB 6830 ! SETUP DISP
6730 B=-2.5*LGT(P*Z)+11 ! 8903A
     ACVOLTS RANGE
6731 IMAGE "1.",DD,"SP19.",3D,"S
     PT0"
6732 OUTPUT 728 USING 6731 ; B.Z
6735 N=0
6740 GOSUB 6800 ! ARROWOFF
6745 A1=48+ATN(4.4*LOG(A/P+.0000
     1))/2 ! CALC DISPLAY POSITI
     ON
6750 GOSUB 6810 ! ARROWON
6755 ENTER 728 ; A
6756 IF A>9000000000 THEN B=0 @
     GOTO 6732 ! OVERLOAD; USE A
     UTORANGE
6760 IF ABS((A-P)/P)>.05 THEN 67
     35
6765 N=N+1 @ IF N<6 THEN 6740
6767 OUTPUT 728 USING 999 ; "1.0
     SP"
6770 GOSUB 5600 ! ET START
6775 B=A
6780 ALPHA
6785 RETURN
6799 !
6800 PEN -1
6805 GOTO 6815
6810 PEN 1
6815 MOVE 13,A1
6820 LABEL ">"
6825 RETURN
6829 !
6830 ! SETUP DISPLAY
6835 GCLEAR @ LDIR 0 @ PEN 1
6840 GRAPH
6845 SCALE 0,100,0,100
6850 YAXIS 20,6.25
6855 YAXIS 19.5,0,43.75,56.25
6860 YAXIS 20.5,0,43.75,56.25
6865 MOVE 26,54 @ LABEL "Adjust
     receiver volume"
6870 MOVE 26,48 @ LABEL "for ind
     ication in"
6875 MOVE 26,42 @ LABEL "center
     range."
6880 RETURN
6899 !
```

Subroutine prompts operator to set receiver audio level

Initialize HP-85 for graphics to be used

Wait for UUT output to stabilize

Measure power

If within 5% of rated power, finished

Stop timer while operator sets level

Set up display graphics

Calculate and set optimum 8903A ac range

This eliminates autoranging, which increases the measurement rate. T0 = free run

In-range counter cleared

Turn old arrow (pointer) off

Calculate new position

Log and arctangent functions give smooth display

Turn on arrow at new position

Read new power

Check for overrange in fixed range mode. If overrange, use autoranging

If error >5%, go around again

Must have 6 readings within 5%

Finished. Turn auto-range back on (1.0 SP)

Start timer again

B = actual set power

Turn off graphics mode

Pen mode = erase

Pen mode = write

Position pen

Write or erase arrow

Clear display; label direction horizontal; pen mode = write; turn on graphics mode

Set scale on X & Y axis

Main scale

Intensify center portion of scale

Operator instructions

Main Program — Data Entry

```
7000 ! ENTER TEST CONDITIONS
7002 CLEAR
7004 IF DATE>100 THEN 7022
7006 DISP "MONTH(1-12)";
7008 INPUT M
7010 DISP "DATE(1-31)";
7012 INPUT D
7014 DISP "TIME(0001-2400)";
7016 INPUT T
7018 SETTIME 60*(T-40*IP(T/100))
     .100*M+D
```

Clear display

Check if timer/calendar has been set

Set timer/calendar

```
7020 CLEAR
7022 PEN 1
7024 GOSUB 5100 ! INIT
7026 DISP "PRESS <END LINE> AFTE
     R"
7028 DISP "EACH DATA ENTRY."
7030 DISP
7032 DIM R$[32],N$[32],L$[51]
7034 DISP "NAME OF TEST OPERATOR
     ";
7036 INPUT N$
7040 DISP "RADIO IDENTIFICATION"
     ;
7042 INPUT R$
7044 DISP "RADIO TYPE(1=AM,2=FM)
     ";
7046 INPUT U
7048 IF U#1 AND U#2 THEN 7044
7050 DISP "RECEIVER FREQUENCY(MH
     z)";
7052 INPUT R
7054 DISP "TRANSMITTER FREQUENCY
     (MHz)";
7056 INPUT T
7058 DISP "RATED AUDIO POWER(WAT
     TS)";
7060 INPUT P
7062 DISP "SPEAKER IMPEDANCE(1-9
     99Ω)";
7064 INPUT Z
7066 IF U=2 THEN 7074
7068 DISP "AM DEPTH FOR TEST(%)"
     ;
7070 INPUT M
7071 Q2=0 @ V=0 @ Y=1 ! NOT USED
      FOR AM
7073 GOTO 7094
7074 DISP "FM DEVIATION FOR TEST
     (kHz)";
7076 INPUT M
7078 M=1000*M ! kHz→Hz
7080 DISP "RECEIVER DE-EMPHASIS(
     μS)";
7082 INPUT V
7084 DISP "RECEIVER SQUELCH FREQ
     UENCY(Hz)-"
7086 DISP "(ENTER 0 FOR NONE)";
7088 INPUT Q2
7090 DISP "TEST STANDARDS"
7091 DISP "(USA/EIA=1,EUR/CEPT=2
     )";
7092 INPUT Y
7093 IF Y#1 AND Y#2 THEN 7091
7094 GOSUB 9100 ! CREATE LOGO
7095 GOSUB 9300 ! WELCOME
7096 K=30 ! dB ATTEN ANT PAD
7097 GOSUB 5500 ! ET CLEAR
7098 BEEP @ PAUSE
7099 !
```

Clear display
Pen mode = write
Initialize instruments and interface
Operator instructions
Strings for radio & operator identification, HP logo
Enter identification and test conditions
AM parameters, if radio is AM
FM parameters, for FM radios
Display identification information
Value of antenna attenuator (may be set as desired)
Clear elapsed timer
Wait for operator to go on

Main Program — Transmitter Tests

```
7100 ! TRANSMITTER TESTS
7101 IMAGE //,7X,"TRANSMITTER TE
     ST",/,7X,"----------------"
     ,3/
7102 IMAGE "Power=",13X,DZ.D," w
     atts"
7103 IMAGE "Frequency=",9X,3D.5D
     ," MHz"
7104 IMAGE "Freq. error=",5X,7D,
     " Hz"
7105 IMAGE "Squelch freq.=",5X,3
     DZ.D," Hz"
7106 IMAGE "Squelch dev.=",6X,DZ
     .DD," kHz"
7107 IMAGE "Mic. sens.=",7X,3DZ
     D," mV"
7108 IMAGE "Distortion=",8X,DZ.D
     D,"%"
7109 IMAGE "Mod. limiting=",4X,D
     DZ.DD,4A
7110 IMAGE "Hum and noise=",5X,3
     D.D," dB"
```

Heading
Images to display results

```
7112 GOSUB 5600 ! ET START                          Start timer
7114 GOSUB 5200 ! XMIT                              59306A/8903A-K85: Transmit Mode
7116 CLEAR                                          Clear display
7118 DISP USING 7101                                Heading
7120 GOSUB 3400 ! AF OFF                            AF level = 0 (for AM carrier power)
7122 GOSUB 2700 ! PEP                               Power measurement
7124 A=A*10^(K/10) ! ATTEN CORR                     Correct for attenuator in signal path
7126 DISP USING 7102 ; A                            Display result
7128 GOSUB 2200 ! CNT                               Count transmitter frequency
7130 DISP USING 7103 ; A*.000001                    Convert to MHz and display
7132 A=A-T*1000000 ! F ERROR                        Calculate error
7134 DISP USING 7104 ; A                            Display error
7136 IF U=1 THEN 7146 ! AM-NO SQ                    Skip squelch test on AM radios
7138 GOSUB 6100 ! SQ TEST                           Check for squelch tone
7140 IF NOT Q THEN 7146 ! SKIP S                    No squelch tone present; don't display
     QUELCH RESULTS
7142 DISP USING 7105 ; Q0                           Display squelch tone frequency
7144 DISP USING 7106 ; Q1/1000                      Display deviation, converted to kHz
7146 A=-32 ! dBv EST SENS                           Estimate for microphone sensitivity (= 25 mv)
7148 B=1 ! kHz AFF                                  1 kHz test frequency
7150 GOSUB 6200 ! MICSENS                           Measure microphone sensitivity
7152 C=A                                            Store value for use by MOD LIM routine
7154 A=1000*10^(A/20) ! dBv→mv                      Convert from dBV to mV
7156 DISP USING 7107 ; A                            Display result
7158 A=V ! μS DEM                                   Set deemphasis filters
7160 B=0
7162 GOSUB 2300 ! DEM
7164 GOSUB 4200 ! DIST                              Measure distortion of transmitter
7166 GOSUB 3300 ! SRC ONLY                          Display 8903A source settings (on 8903A)
7168 DISP USING 7108 ; A                            Display distortion result
7170 A=C                                            Set A to mic. sensitivity for MOD LIM
7172 GOSUB 6300 ! MODLIM                            Measure modulation limiting
7174 IF U=1 THEN DISP USING 7109                    Display result in % or kHz for AM or FM
       ; B," %"
7176 IF U=2 THEN DISP USING 7109
       ; B/1000," kHz"
7178 GOSUB 2900 ! FIX MOD RANGE                     Stop autoranging when measuring S/N ratio
7180 GOSUB 4400 ! S/N                               Measure residual hum and noise with 8903A
7182 DISP USING 7110 ; -A                           Display result
7184 GOSUB 5700 ! ET STOP                           Stop timer
7186 GOSUB 5300 ! RCV                               59306A/8903A-K85: Receive mode
7188 BEEP @ PAUSE                                   Wait for operator
7199 !
```

Main Program — Receiver Tests

```
7200 ! RECEIVER TESTS
7201 IMAGE //,8X,"RECEIVER TEST"                    Heading
     ,/,8X,"-------------",3/
7202 IMAGE "10dB S/N sens.=",3X,                    Images to display results
     DDZ.DD," μV"
7203 IMAGE DD,"dB SINAD sens.=",
     X,DDZ.DD," μV"
7204 IMAGE "Audio power=",6X,DDZ
     .3D," watts"
7205 IMAGE "Signal-to-noise=",4X
     ,DD.D," dB"
7206 IMAGE "Distortion=",8X,DZ.D
     D,"%"
7208 GOSUB 5600 ! ET START                          Start timer
7210 CLEAR                                          Clear display
7212 DISP USING 7201                                Heading
7214 A=-113 ! dBm EST                               Initial sensitivity estimate (= 0.5 μV)
7216 GOSUB 6500 ! RXSENS                            Measure receiver senstivity
7218 A=1000000*SQR(.05*10^(A/10)                    Convert dBm to μV
     ) ! dBm→μv
7220 IF U=1 THEN DISP USING 7202                    Display result using appropriate image
       ; A
7222 IF U=2 THEN DISP USING 7203
       ; 4+8*Y,A
7224 A=-47 ! -47dBm=1000μv                          Reference level for audio measurements= 1000 μV
7226 GOSUB 1100 ! RFL                               Set level
7228 GOSUB 4500 ! WATTS                             Measure audio power
7230 DISP USING 7204 ; A                            Display result
7231 IF Q2 THEN 7236 ! CAN'T DO                     Skip S/N if 8903A source is used for a squelch tone
     SN WITHOUT 8903 SOURCE
7232 GOSUB 4400 ! S/N                               Measure S/N ratio
7234 DISP USING 7205 ; A                            Display result
7236 GOSUB 4200 ! DIST                              Measure receiver audio distortion
7238 DISP USING 7206 ; A                            Display result
7240 GOSUB 5700 ! ET STOP                           Stop timer
7242 BEEP @ PAUSE                                   Wait for operator
7299 !
```

Main Program — Audio Response

```
7300 ! RECEIVER AUDIO RESPONSE
7301 IF Q2 THEN 7400
7302 GCLEAR @ GRAPH
7304 LDIR 0
7306 SCALE -500,3400,-14,10
7308 MOVE 500,9 @ LABEL "AUDIO R
     ESPONSE"
7310 XAXIS 0,250,0,3000
7312 YAXIS 0,1,-10,10
7314 FOR A=0 TO 3 ! FREQ AXIS
7316 MOVE 1000*A,-12
7318 LABEL VAL$(A)
7320 NEXT A
7322 MOVE 750,-14 @ LABEL "FREQU
     ENCY kHz"
7324 FOR A=-10 TO 10 STEP 5
7326 ! LEVEL AXIS
7328 MOVE -170-120*((A<0)+(ABS(A
     )=10)),MIN(A-.5,9.1)
7330 LABEL VAL$(A)
7332 NEXT A
7334 LDIR 90
7336 MOVE -340,-5.2 @ LABEL "RES
     PONSE dB"
7338 MOVE 2630,10 @ BPLOT L$,3 !
      HP LOGO
7340 ! PLOT GRAPH
7342 GOSUB 5600 ! ET START
7344 GOSUB 4100 ! ACV
7346 OUTPUT 728 USING 999 ; "FR1
     KZH0T3,R1LG" ! SET dB RATIO
     ,HP FLT OFF
7348 B=10*LGT(1+(.002*PI*V)^2) !
      CORR FACTOR FOR DEEMPHASIS
      @ 1kHz
7350 FOR F=200 TO 3000 STEP 200
7352 A=F/1000
7354 GOSUB 3100 ! AFF
7356 GOSUB 4600 ! DATA
7358 A=A-B+10*LGT(1+(.000002*PI*
     F*V)^2) ! CORR FOR DEEMPHAS
     IS
7360 PLOT F,A @ PEN 1
7362 NEXT F
7364 GOSUB 5700 ! ET STOP
7366 LDIR 0
7368 MOVE 1100,6 @ LABEL "*TEST
     COMPLETE*"
7370 BEEP @ PAUSE
7399 !
```

Skip test if 8903A source required for squelch
Select graphics mode; clear display
Label horizontally
Scale allows frequency & response to be plotted without conversions
Heading
Main axes
Label frequency axis
Label response axis
Display HP Logo
Start timer
Set 1 kHz log reference for ac level measurements. 8903A will calculate response error internally
Deemphasis value at reference frequency
Frequency steps
Set source frequency in kHz
Measure response (in dB re 1 kHz)
Correct for receiver ideal deemphasis curve
Plot response error
Next frequency
Stop timer
Label horizontal
Operator message
Wait for operator

Main Program

```
7400 ! END OF TEST
7402 IMAGE //,"Elapsed test time
     :",4D.D," seconds",7/
7404 PRINT USING 7402 ; T1
7406 GOSUB 9200 ! TITLE BLOCK
7408 PAUSE
7410 GOSUB 5500 ! ET CLEAR
7412 GOTO 7100 ! RETEST
7999 END
```

Print total time for measurements
Display HP Title Block
Wait for operator
Clear timer
Restart test

Graphics Subroutines

Lines 9000–9500.
These subroutines create the HP Logo and Title Block which are visible in the display at various times.

November, 1980
5952-8241
Printed in U.S.A.